POSITIVITY OF TORIC VECTOR BUNDLES

ANDREAS HOCHENEGGER

Abstract. In this note, we provide an overview of the *PositiviyToricBundles* package for *Macaulay2*. This package implements algorithms to determine whether a toric vector bundle on a smooth projective variety is nef, (very) ample or globally generated. The theoretical bases is given by [3] and [5], wheras the implementation depends on the *ToricVectorBundles* package described in [1].

1. Introduction

Let $X = \mathrm{TV}(\Sigma)$ be a toric variety of dimension n over an algebraically closed field $\Bbbk$ of characteristic zero. A *toric vector bundle* $\mathcal{E}$ is a vector bundle on X which is equivariant with respect to the torus $T = (\Bbbk^*)^n$ acting on X. In [2], Alexander Klyachko gave a description of such bundles in terms of filtrations. The idea is to consider the fiber $E = \mathcal{E}_x$ over a point $x \in T \subset X$, which is a $\Bbbk$-vector space of dimension $\mathrm{rk}(\mathcal{E})$. Tracking this fibre of $\mathcal{E}$ along one-parameter subgroups corresponding to the rays ρ of the fan Σ, yields filtrations $(E^\rho(i))_{i\in\mathbb{Z}}$:

$$E \supseteq \quad \cdots \supset E^\rho(i-1) \supset E^\rho(i) \supset E^\rho(i+1) \supset \cdots \quad \supseteq 0.$$

Conversely, given filtrations for each ray that satsify a compatibility condition, one can construct a toric vector bundle. The main theorem of [2] is that there is an equivalence between such filtrations and the category of toric vector bundles.

This framework of filtrations make toric vector bundles very accessible to explicit computations. For example, this description allows the *ToricVectorBundles* package for *Macaulay2* to perform several constructions on toric vector bundles (tensor product, symmetric and exterior power, ...) and compute their cohomology, see [1]. Additionally this description also enters the picture in results about the positivity of toric vector bundles.

1.1. **Nef and ample toric vector bundles.** Recall that a vector bundle $\mathcal{E}$ on an arbitrary variety X is *nef* or *(very) ample* if the line bundle $\mathcal{O}$ on the projective bundle $\mathbb{P}(\mathcal{E})$ is nef or (very) ample.

For toric vector bundles, the situation becomes much easier thanks to the following result.

Theorem 1.1 ([3, Thm. 2.1])**.** *Let $\mathcal{E}$ be a toric vector bundle on a complete toric variety X. Then $\mathcal{E}$ is nef or ample if and only if its restriction to every torus invariant curve is nef or ample, respectively.*

Note that every invariant curve C of a complete toric variety is isomorphic to $\mathbb{P}^1$, so $\mathcal{E}|_C$ splits into a direct sum $\mathcal{O}_{\mathbb{P}^1}(a_1) \oplus \cdots \oplus \mathcal{O}_{\mathbb{P}^1}(a_{\mathrm{rk}(\mathcal{E})})$. Hence, we only need to check whether all these numbers a_i are non-negative or positive to conclude that $\mathcal{E}$ is nef or ample, respectively.

1.2. **Globally generated and very ample toric vector bundles.** It is also possible to decide whether a toric vector bundle is globally generated or very ample. But the answer is not as simple as for nef and ample.

In [5], the notion of a *parliament of polytopes* for a toric vector bundle on a smooth complete toric variety is introduced. This notion generalises the polytope of global sections of a line bundle. The second ingredient is the *toric Chern character* of a toric vector bundle from [4].

The crucial result [5, Thm. 6.2] is that the parliament of polytopes and the toric Chern character tell us when a toric vector bundle $\mathcal{E}$ *separates ℓ-jets*, i.e. for every point $x \in X$ the map

$$H^0(X, \mathcal{E}) \to H^0(X, \mathcal{E} \otimes_{\mathcal{O}_X} \mathcal{O}_X/\mathfrak{m}_x^{\ell+1})$$

is surjective (where $\mathfrak{m}_x$ is the maximal ideal sheaf of x). Note that for $\ell = 0$, we recover the definition of globally generated.

In general, being separated by 1-jets and very ampleness do not coincide, but by [5, Thm. 6.6] the toric situation is much nicer.

Theorem 1.2. *Let $\mathcal{E}$ be a toric vector bundle on a smooth complete toric variety X. Then $\mathcal{E}$ is very ample if and only if it separates 1-jets.*

2. The package *PositivityToricBundles*

This package can check whether a toric vector bundle $\mathcal{E}$ on a complete toric variety X is

(1) nef or ample;
(2) globally generated or very ample (if additionally X is smooth).

As input, the methods of this package always take toric vector bundles in Klyachko's description, i.e. they are of the type `ToricVectorBundleKlyachko` from the *ToricVectorBundles* package. Although it is very convienent to have all the methods there available, this also means that we follow the sign convention from there:

- the fan associated to a polytope will be generated by inner normals,
- the filtrations for describing a toric vector bundle are *increasing*.

The first choice is very common in toric geometry, but is the opposite as the one used in invariant theory. Unfortunately for us, in [3], [4] and [5] the opposite choice is taken. The second choice is in contrast to the one taken in [2]. Obviously, both choices do not change the mathematical content, but may result in cumbersome tracking of the right signs.

2.1. **Nef and ample.** As explained in Section 1.1, we can look at the restrictions of $\mathcal{E}$ to the invariant curves $C \cong \mathbb{P}^1$ of X. This is implemented as the method `restrictToInvCurves` following the algorithm in [5, §5], on this method the simple checks `isNef` and `isAmple` are built.

Example 2.1. We consider the tangent bundle $\mathcal{T}$ on $\mathbb{P}^2$.

```
i1: needsPackage "PositivityToricBundles";
i2 : E = tangentBundle(projectiveSpaceFan 2)
o2 = {dimension of the variety => 2 }
      number of affine charts => 3
      number of rays => 3
      rank of the vector bundle => 2
o2 : ToricVectorBundleKlyachko
```

Now we let the package compute the restrictions.

```
i3 : restrictToInvCurves E
o3 = HashTable{| -1 | => {2, 1}}
               | -1 |
               | 0 | => {2, 1}
               | 1 |
               | 1 | => {1, 2}
               | 0 |
o3 : HashTable
```

This hash table associates to each cone of codimension 1 (here: a ray) of the fan a list of integers (a_1^ρ, a_2^ρ). As cones of codimension 1 corresponds to the invariant curves C of the toric variety, this result tells us that $\mathcal{T}|_C \cong \mathcal{O}_{\mathbb{P}^1}(2) \oplus \mathcal{O}_{\mathbb{P}^1}(1)$ for all these curves. From this we see that the bundle is nef and even ample:

```
i4 : isNef E
o4 = true
i5 : isAmple E
o5 = true
```

2.2. **Globally generated and very ample.** In contrast to check whether a toric vector bundle on a smooth toric variety is globally generated or very ample, we can compute whether $\mathcal{E}$ separates 0- or 1-jets, as explained in Section 1.2. For this, the method `separatesJets` is implemented, following the characterisation in [5, Thm. 6.2]. On top of this, the methods `isGloballyGenerated` and `isVeryAmple` are just simple checks.

Example 2.2. We continue the example of the tangent bundle $\mathcal{T}$ on $\mathbb{P}^2$.

```
i6 : separatesJets E
o6 = 1
```

This tells us that $\mathcal{T}$ separates 1-jets. So it is globally generated and very ample:

```
i7 : isGloballyGenerated E
o7 = true
i7 : isVeryAmple E
o7 = true
```

2.3. **Further example.** We end with a more complicated example, namely a toric vector bundle $\mathcal{E}$ on $\mathbb{P}^2$ which is ample and globally generated, but not very ample. This is [5, Ex. 6.4].

First we set up the bundle $\mathcal{E}$.

```
i1 : needsPackage "PositivityToricBundles";
i2 : E = toricVectorBundle(3, projectiveSpaceFan 2);
```

```
i3 : rays E
o3 = {| -1 |, | 0 |, | 1 |}
      | -1 |  | 1 |  | 0 |
o3 : List
```

The *ToricVectorBundles* package chooses here an order on the rays, which we use in the following to enter the filtrations.

```
i4 : E = addBase( E, {
          matrix{{ 1, 0, 0},
                 {-1, 1, 0},
                 { 0,-1, 1}}, -- for (-1,-1)
          matrix{{0,0,1},
                 {0,1,0},
                 {1,0,0}},    -- for (0,1)
          matrix{{1,0,0},
                 {0,1,0},
                 {0,0,1}}     -- for (1,0)
         });
i5 : E = addFiltration( E, {
          matrix{{-4,-3,-1}},  -- for (-1,-1)
          matrix{{-2,0,2}},    -- for (0,1)
          matrix{{-2,1,2}}     -- for (1,0)
         });
```

Here some explanation is needed. The columns of the matrices in `i4` are the bases vectors which appear in the filtrations, and the entries in the row matrices in `i5` tell us when these vectors appear in the filtration. Note that the filtrations in [5, Ex. 6.4] have the opposite signs. Once the bundle is set up, the checks become easy:

```
i6 : isAmple E
o6 = true
i7 : isGloballyGenerated E
o7 = true
i8 : isVeryAmple E
o8 = false
```

As a bonus, we can also let the package create a picture for us, which uses TikZ.

```
i9 : drawParliament2Dtikz (E,"RJS-Ex6.4.tikz")
```

The result of `drawParliament2Dtikz` is the following.



Again, if we compare this with [5, Fig. 5], the picture is switched due to different sign conventions. The gray triangles in the picture are the polytopes of the parliament, the light blue dots indicate the degrees of the global sections, whereas the yellow circles are the components of the toric Chern character. As this picture involves a degenerate polytope, we use the methods `parliament` and `toricChernCharacter` to get direct access to the parliament of polytopes and the toric Chern character:

```
i10 : apply( values parliament E, vertices)

o10 = {| 2  3  2  |, | 1 |, | 1 2 1 |, | 2 3 2 |, | -2 -1 -2 |}
       | -2 -2 -1 |  | 0 |  | 2 2 3 |  | 0 0 1 |  | 2  2  3  |
o10 : List
```

Here we have also used `vertices` to get the vertices of these polytopes.

```
i11 : toricChernCharacter E
o11 = HashTable{| -1 0 | => {| 2 |, | 3 |, | 3  |}}
                | -1 1 |     | 2 |  | 0 |  | -2 |
                | 1 -1 | => {| -2 |, | 1 |, | 2 |}
                | 0 -1 |     | 3  |  | 3 |  | 1 |
                | 1 0 | => {| -2 |, | 2  |, | 1 |}
                | 0 1 |     | 2  |  | -2 |  | 0 |
o11 : HashTable
```

From this we see very directly that the bundle separates 0-jets but not 1-jets, using the characteristion from [5, Thm 6.2]: each component of the toric Chern character is a vertex of a polytope of the parliament (so it separates 0-jets), but the adjacent edges do not have length at least 1 (necessary to separate 1-jets), as one polytope is just a point.

References

[1] René Birkner, Nathan Ilten, Lars Petersen, *Computations with equivariant toric vector bundles*, JSAG **2** (2010), 11–14.

[2] Alexander Klyachko, *Equivariant vector bundles on toral varieties*, Izv. Akad. Nauk SSSR Ser. Math. **5** (1989), 1001–1039 (in Russian), Math. USSR Izv. **35** (1990), 337–375 (in English).

[3] Milena Hering, Mircea Mustaţă, Sam Payne, *Positivity Properties of toric vector bundles*, Ann. Inst. Fourier (Grenoble) **60** (2010), 607–640, also arXiv:0805.4035.
[4] Sam Payne, *Moduli of toric vector bundles*, Compositio Math. **144** (2008), 1199–1213, also arXiv:0705.0410.
[5] Sandra Di Rocco, Kelly Jabbusch, Gregory Smith, *Toric vector bundles and parliaments of polytopes*, Trans. AMS **370** (2018), 7715–7741, also arXiv:1409.3109.

Scuola Normale Superiore, Piazza dei Cavalieri 7, 56126 Pisa, Italy
Email address: andreas.hochenegger@sns.it